<commit_before>function! deleft#closing#SimpleDeleft()
  if !exists('b:deleft_closing_pattern')
    let b:deleft_closing_pattern = '.'
  end

  let start_lineno = line('.')
  let start_indent = indent(start_lineno)

  let current_lineno = nextnonblank(start_lineno + 1)
  let found_end = 0

  while current_lineno <= line('$')
    if indent(current_lineno) > indent(start_lineno)
      let current_lineno = nextnonblank(current_lineno + 1)
      continue
    endif

    if indent(current_lineno) == indent(start_lineno) &&
          \ getline(current_lineno) =~ b:deleft_closing_pattern
      let found_end = 1
      break
    endif

    let current_lineno = nextnonblank(current_lineno + 1)
  endwhile

  let end_lineno = current_lineno

  if end_lineno - start_lineno > 1
    silent exe (start_lineno + 1).','.(end_lineno - 1).'<'

    if found_end
      " then the end line is the block-closer, delete it
      silent exe end_lineno.'delete _'
    endif
  endif

  silent exe start_lineno.'delete'
  echo
endfunction<commit_msg>Try fixing an issue with simple delefting<commit_after>function! deleft#closing#SimpleDeleft()
  if !exists('b:deleft_closing_pattern')
    let b:deleft_closing_pattern = '.'
  end

  let start_lineno = line('.')
  let start_indent = indent(start_lineno)

  let current_lineno = nextnonblank(start_lineno + 1)
  let found_end = 0

  while current_lineno <= line('$')
    if indent(current_lineno) > indent(start_lineno)
      let current_lineno = nextnonblank(current_lineno + 1)
      continue
    endif

    if indent(current_lineno) == indent(start_lineno) &&
          \ getline(current_lineno) =~ b:deleft_closing_pattern
      let found_end = 1
      break
    endif

    if indent(current_lineno) < indent(start_lineno)
      " haven't found an end, but indentation is off, so let's just delete
      " what we have
      silent exe start_lineno.'delete'
      return
    endif

    let current_lineno = nextnonblank(current_lineno + 1)
  endwhile

  let end_lineno = current_lineno

  if end_lineno - start_lineno > 1
    silent exe (start_lineno + 1).','.(end_lineno - 1).'<'

    if found_end
      " then the end line is the block-closer, delete it
      silent exe end_lineno.'delete _'
    endif
  endif

  silent exe start_lineno.'delete'
endfunction
{'fraction_non_alphanumeric': 0.08888888888888889, 'fraction_numerical': 0.006971677559912854, 'mean_word_length': 3.3484848484848486, 'pattern_counts': {'":': 0, '<': 8, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4793849', 'n_tokens_mistral': 812, 'n_tokens_neox': 768, 'n_words': 187}